
The Expert's Guide to Cross-Tabs in Crystal Reports 
Addendum for Grid functions and features. 

Adding rows or columns that do fixed calculations: 
 

All Crystal versions starting with v12 (CR 2008) allow you to insert an individual row or column 

anywhere in the Cross-tab.  This column can do calculations based on the other rows or columns 

in the cross-tab. The calculation can refer to rows or columns that have a specific row or column 

name (like Sales) or to columns that have a relative position to the calculation (like 2 columns to 

the left).  

 

For example, if you have rows for Gross Sales and Cost of Goods Sold you could insert a row for 

Net Sales that takes the difference between these two rows.  If you specify these rows by name, 

then they are used in the calculation even if these rows are moved up and down by changes in the 

other rows of the cross-tab.  Or, lets say you have 12 monthly columns.  You could insert a new 

column so that it compares the value of the column to its left to the value of the column that is 

three columns to the left (the prior month).  

 

First let’s demonstrate columns specified by name, since these are almost automatic to 

implement.  Once they are created you can see the features that make them work.  

 

Exercise 1 
 

Open starter report that I have included.  Notice that there are two countries and each is 

subdivided into six shipping methods.  I want to add a row that adds together FedEx and UPS 

and gives me a new row labeled "Express Delivery" - but only within the US.   

 

Right-click on the "FedEx" row heading within the USA group and select "Calculated Member" 

and within that "Select USA -> FedEx as first value".  Click OK on the message.  At this point 

nothing in the Cross-tab changes but the Cross-tab will remember that choice.  Now right-click 

on the UPS heading (again in USA) and select "Calculated Member" and notice that there are 

four new choices.  These allow you to use the selected values for adding, subtracting, 

multiplying or dividing.  Select the "Sum" of the two values and note the following changes.  

 

1) A new row appears titled simply "Sum".  We will change that later.  

2) This row has the sum of the two selected rows added together.  Note that this row only exists 

in the USA group and doesn't appear in the Canada group. Note also that this row does not 

change the values in the grand total row since these numbers are already included.  

 

Now to see what is happening.  Right-click on the word SUM and select "Calculated Member" 

then "Edit Header Formula".  What you will see is a string formula that you can use to control 

what is printer here.  Replace the word "Sum" with "All Express" and then Save and Close.  Note 

the change in the header formula.  

 

Right-click on the words "All Express" and select "Calculated Member" then "Edit Insertion 

Formula".  Remove the middle part that says: 



 

and GridRowColumnValue("Customer.Country") = "USA"  

 

Note that the "All Express" line now appears in both groups, but in both cases it is showing the 

total from the USA group.  It is currently using fixed positions rather than relative positions.   

 

Right-Click on the upper summary value (the Count value) just to the right of the "All Express" 

label, under UPS (either country).  Select "Calculated Member" and "Edit Calculation Formula".  

Make the following two changes to make the formula relative to the current Country group: 

 

replace: 

 GetRowPathIndexOf("USA", "FedEx") 

with: 

 GetIndexOf("Orders.Ship Via","FedEx") 

 

replace: 

 GetRowPathIndexOf("USA", "UPS") 

with: 

 GetIndexOf("Orders.Ship Via","UPS") 

 

 

The "GetRowPathIndexOF" function specifies both levels of this two-level cross-tab, which 

returns one specific row no matter where you are.   In contrast the "GetIndexOF" function 

specifies only one level.  When you leave a level unspecified in a multi-level cross-tab, Crystal 

will default to the current level.  This returns a value relative to the cell doing the calculation. So 

by specifying "FedEx" in the last example I get the Fedex value within the current country, not 

the one from the literal country “USA”.   

 

Once you have this calculation working in the first summary fields (the Count value) you can 

copy the entire formula and replace the same formula in the lower summary value (the Sum).  

Because the formula uses “CurrentSummaryIndex” the lower summary calculation will 

automatically use the values from the bottom of the cells while the upper formula uses the values 

from the top of the cells.  

 

At this point your cross-tab should look like the one in the attached Starter Exercise 1.rpt. 

 



A few words of caution: 

1) If you specify a value you have to make an exact and even case sensitive match to the literal 

value in the cross-tab.  If  you misspell the literal or if the specified value doesn't exist in one of 

the groups then the entire report stops with an error message saying that the value is "out of 

range". You can prevent this by checking for existence of the value first, like this: 

 

 if GetIndexOf("Orders.Ship Via","FedEx") > 0 then ... 

 

This works because when there is no Fedex row the index will be -1, and the formula can skip 

the calculation. 

 

 

2) When a new row is first created it will have the same number of summary rows as the existing 

rows. But if you then add another summary field to the cross-tab (after adding the new row) the 

new row will have a new empty space where the new summary positions would go.  It will not 

add new summary positions in that space and I could not find a way to force it to create new 

summary positions in the empty space.  The only option seems to be deleting the calculated row 

and adding it back in again once you have added the needed number of summary fields.   

 

So calculated rows can have fewer summary positions than the rest of the cross-tab, but these 

rows will still have the same depth as the other rows.  The extra space will be blank.  It also 

means that deleting a summary field within the cross-tab expert will delete the corresponding 

summary positions in all calculated rows.   

 

3) Calculated values in a manual row must have the same data type as the corresponding 

summary field in the regular cells.  They will also share the same formatting attributes.  If you 

need the manual row to return values of a different data type or in a different format then you 

must use the “Display String” property to generate and format the values.   

 

One surprising related behavior.  If the first summary field in your cross-tab is a Count or 

Distinct Count than any calculating you do that is in the corresponding first position will be 

forced to be an integer value.  If the actual calculation doesn't return an integer Crystal will 

truncate (not round) the calculated value so it is an integer.  If you want your value rounded 

instead of truncated then you can use the round function within the row calculation (i.e. before 

Crystal will truncate it).  But to get decimals you would have to use a display string.  

 

4) In my experiments I found that if I made many changes to the same Cross-tab I could corrupt 

the internal index numbering.  This caused the row and column references to get reversed, 

generating odd errors regarding the data types of the fields.  The only solution I found when this 

happened was to recreate the cross-tab groups or in some cases to recreate the entire cross-tab.  

 

 



Relative Calculations: 
 
To illustrate relative calculations we are going to add a column after each month to show the 

variance from the value of the month before.  I don’t want to see a variance column after the first 

month because there will be no prior month to compare.  This also allows me to show you  how 

to make an exception to the rule.   

 

Exercise 2 
 
You can start with either the original Starter report or continue with the one from the last 

exercise.  Right-click on the column heading for the second month (August), and select 

"Calculated Member > Insert Column".  Click OK on the message.  This will create a new 

column to the right of that column.  The new column will have a blank heading and all zeros.   

 

Right click on the empty heading cell and select "Calculated Member > Edit Header Formula".  

Put a literal label there that is "Variance".   

 

Now Right click on the cell below the heading cell (first summary) and select "Calculated 

Member > Edit Calculation Formula". Replace this formula with the function 

"CurrentRowIndex". This will show you the row number for each row and you will see that the 

first row is zero (not one).  After that they are numbered consecutively.  

 

Now Right click again on the cell below the heading cell and select "Calculated Member > Edit 

Calculation Formula". Change it to the function "CurrentColumnIndex".  Save and close and 

note that the variance column will display its column number as two.  The new column is 

number two because it is the third column (0, 1, 2).  Again, the first summary column is number 

zero. The same is true for the summary index which counts the number of summary calculations 

in each cell.  Since every cell in this cross-tab has 2 summaries they would be numbered 0 and 1.  

 

Right click on the top cell and select "Calculated Member > Edit Insertion Formula.  Change that 

formula to read: 

 CurrentColumnIndex > 0 

 

We are saying to insert this new column to the right of every column except column 0 (the first 

column). We skip the first column because there is no prior month for comparison to the first 

column.  When you save you should see the Variance column now appear after all months to the 

right, but not after the first column.  

 

Now Right click again on the first summary field below the heading and select "Calculated 

Member > Edit Calculation Formula".  Put in the following formula: 

 

GridValueAt (CurrentRowIndex, CurrentColumnIndex -1  , CurrentSummaryIndex )  - 

  if CurrentColumnIndex = 2 

 then GridValueAt (CurrentRowIndex, CurrentColumnIndex -2 , CurrentSummaryIndex )  

 else GridValueAt (CurrentRowIndex,CurrentColumnIndex -3 , CurrentSummaryIndex ) 

 



This looks complex, but isn’t so bad when you break it down.  The first line of the formula 

specifies the value from the current month. It says to use the current row but use the column one 

place to the left of the current column (index minus one).   Then we give the coordinates for the 

prior month.  We need an  IF-THEN statement because the prior month is two columns to the left 

in August but it is three columns to the left in all of the other months.  So you see that the 

CurrentColumnIndex has minus 2 and minus 3 in those lines.  

 

You can copy the same formula to the second summary field.  Because it uses the function 

CurrentSummaryIndex it knows that the top or first summary (index 0) corresponds to the top 

summary in each cell, while the bottom one (index 1) corresponds to the bottom of each cell. 

 

At this point your cross-tab should correspond to Starter Exercise 2.rpt.  

 

  



List of Grid Functions: 
The exercises above illustrate what I consider to be the most useful functions in most situations.  

Below are all of the functions available with a brief description of what they do.   Those without 

parentheses do not need any arguments. Those with parentheses require the arguments described 

in the example. For more information on these, including a full syntax example, refer to Crystal 

Reports Help.  

 

GridRowColumnValue () -  

 Returns the Literal of the current Row/Column for comparison. 

 Use a Row field for Inserted Rows, Column field for Inserted Columns 

i.e. if GridRowColumnValue ("Customer.Country") = "USA" then… 

 

GridValueAt() – What is the summary value for a given set of coordinates 

GridLabelAt ()- What label is the Nth value for a given row/column field 

GetNumRows - how many rows are in the output (including total row) 

GetNumColumns - how many columns are in the output (including total column) 

 

GetNumColumnGroups - How many column levels exist (fields in the Column box) 

GetNumRowGroups - How many row levels exist (fields in the Row box) 

GetNumSummaries - How many summary fields exist (fields in the Summarized Fields box) 

 

CurrentRowIndex – What is the current row of output in zero based index 

CurrentColumnIndex - What is the current column of output in zero based index 

CurrentSummaryIndex - What is the current summary row of output in zero based index 

 

GetTotalValueFor () 

What is the total of the current Row or Column.   

If you inserted a Row use the Row field name to get a value from the Total Row of this Column 

If you inserted a Column use the Column field name to get values from the Total Column of this 

Row 

 

GetIndexOf ("Customer.Country","USA" ) 

Find the row number or column number of this cross-tab label 

 

GetRowPathIndexOf ()  /  GetColumnPathIndexOf () 

Give a literal or heirarchy of literals and get the corresponding row or column number:  

GetRowPathIndexOf ('USA', 'FedEx' ) 

 

GetColumnGroupIndexOf ()   /  GetRowGroupIndexOf () 

What level is this group field?  

 



Functions not illustrated: 

GridLabelAt  

GetNumRows  

GetNumColumns 

GetNumColumnGroups  

GetNumRowGroups  

GetNumSummaries  

GetTotalValueFor  

GetColumnGroupIndexOf () 

GetRowGroupIndexOf () 

 

Right click on Total label / select "Calculated Member > Insert Child 

Right click on a summary and select "Insert Embedded Summary" 

 

Display string for subtotal row? 

if CurrentColumnIndex in [2,4,6,8,10] 

then totext (CurrentFieldValue+.5,1,'') & " %" 

else totext (CurrentFieldValue,0,'') 

 


